Paper 172-30

Using SAS®9 ODS Features to Present Table and Graph Data in an Adobe PDF File

Lori S. Parsons, Ovation Research Group, Seattle, Washington

ABSTRACT

An Adobe® Portable Document Format (PDF) file is an ideal way to create and share data across different platforms and across the Internet. PDF is a universal file format that preserves the fonts, images, graphics, and layout of any source document, regardless of the application and platform used to create it. A PDF file is also an especially good format for organizing large amounts of data. The SAS® Output Delivery System (ODS) allows output to be printed directly to a PDF file. Both SAS® procedure output and SAS/GRAPH® output can be combined in a single PDF file. PDF bookmarks can be created to help the viewer navigate the document. Starting in the SAS®9 release, ODS LAYOUT and ODS REGION can be used to define areas on the page where output is placed.

This paper will describe how ODS can be used to produce, organize, and display output in a PDF file. New ODS options available in the SAS®9 release, such as ODS LAYOUT and ODS REGION, will be discussed. Examples of combined table and graph data will be given. The code presented in this paper was run with SAS Version 9.1.2 on Windows XP Professional. The PDF files are viewed with Adobe® Reader® 6.0. The reader should have a basic understanding of macro programming and ODS code.

INTRODUCTION

It has become increasingly popular to produce reports and distribute them via the Internet. Producing organized reports in files that are easy to navigate and appear in the same format as they were created is especially important. Adobe® PDF files have become the industry standard for sharing files across platforms. We have had the ability to create PDF files with custom bookmarks directly from SAS ODS code since version 8. With the addition of ODS LAYOUT and ODS REGION in SAS®9 (though still experimental), we can now place tables and graphs on a single page exactly where we want them.

In the examples that follow, the pertinent SAS code has been provided. The complete SAS code can be found in the appendices of this paper. The code to create the example dataset is in Appendix 1. The code to create the PDF file in example 1 is in Appendix 2, the code to create the PDF file in example 2 is in Appendix 3, and the code to create the PDF file in example 3 is in Appendix 4.

METHODS

The Adobe Portable Document Format (PDF)

PDF is a universal file format that preserves the fonts, images, graphics, and layout of any source document, regardless of the application and platform used to create it. Adobe® PDF files are compact and complete, and can be shared, viewed, and printed by anyone with free Adobe® Reader® software.

Creating a PDF file in SAS

The ODS code to create a basic PDF file is very simple. The code is:

```
ODS PDF file = "the path name goes here/filename.PDF";
   /* SAS procedures go here */
   run;
   quit;
ODS PDF close;
```

ODS PDF FILE = opens the destination and file; ODS PDF CLOSE closes the destination and file. The RUN and QUIT commands before the ODS PDF close statement are required. Otherwise, the output from the last procedure may not be included in the document.

Paper 172-30

ODS PDF Parameters

There are several parameters that provide general information to the final document. These options include: AUTHOR = " ", KEYWORDS = " ", SUBJECT = " ", and TITLE = " ". The following code is an example of how to create a PDF file and include these options.

```
ODS PDF file="C:\Projects\SUGI_30\Output\SUGI30_1.PDF"
   author = 'Lori Parsons'
   keywords = 'Regional Discharge Medications Annual'
   subject = 'Process of Care by Region'
   title = 'Hospital Discharge Medications by Region and Year of Admission';
ODS PDF close;
```

The information can then be viewed in the *Description* section of the *Document Properties* dialog when the PDF is viewed in Acrobat Reader. Figure 1 shows the results.

Figure 1:



Styles and Templates

ODS allows you to choose the look of the output with the STYLE = option. Style definitions define the background color, table borders, color scheme, fonts, sizes, and color of text. There are over 35 styles definitions provided in SAS®9. If you cannot find a sytle that is exactly what you want, you can create your own with PROC TEMPLATE. Many very good papers have been written that describe using styles and PROC TEMPLATE, so that will not be discussed in detail here. For the examples in this paper, the default style is used. The default style for the printer family of ODS output is *Printer*. To change the default style from *Printer* to *Statistical*, use the following code:

```
ODS PDF file="C:\Projects\SUGI_30\Output\SUGI30_1.PDF"
   author = 'Lori Parsons'
   keywords = 'Regional Discharge Medications Annual'
   subject = 'Process of Care by Region'
   title = 'Hospital Discharge Medications by Region and Year of Admission'
   style = Statistical;
ODS PDF close;
```

Paper 172-30

File Size and Compression

Prior to SAS®9, it was not possible to compress a file created by ODS PDF. This can pose a problem, especially if a very large file is to be sent across the Internet. A new feature in SAS®9 is the ability to compress a PDF file created with ODS with the COMPRESS = option .

The syntax is:

```
ODS PDF file="C:\Projects\SUGI_30\Output\SUGI30_1.PDF"
   author = 'Lori Parsons'
   keywords = 'Regional Discharge Medications Annual'
   subject = 'Process of Care by Region'
   title = 'Hospital Discharge Medications by Region and Year of Admission'
   compress = n;
ODS PDF close;
```

The valid range for the value parameter **n** is 0 to 9 where 0 means no compression and 9 means maximum compression. The default is 6. Figure 2 shows the file sizes of the same output when a compression of 0, 1, 3, 6 and 9 is used.

Figure 2:



Name	Size	Type
Compress_EQ_0.PDF	684 KB	Adobe Acrobat Document
Compress_EQ_1.PDF	170 KB	Adobe Acrobat Document
Compress_EQ_3.PDF	161 KB	Adobe Acrobat Document
Compress_EQ_6.PDF	139 KB	Adobe Acrobat Document
Compress_EQ_9.PDF	139 KB	Adobe Acrobat Document

PDF Bookmarks in SAS

Bookmarks in a PDF file are similar to a table of contents in a HTML file. The bookmarks link to the corresponding section of the document. This is a very useful feature, especially while navigating through large amounts of data. By default, SAS will automatically create bookmarks for the PDF file by using the internal procedure labels. It is very ease to modify the default labels with the ODS PROCLABEL **=** statement. The code is:

```
ODS PDF file="C:\Projects\SUGI_30\Output\SUGI30_1.PDF"
   author = 'Lori Parsons'
   keywords = 'Regional Discharge Medications Annual'
   subject = 'Process of Care by Region'
   title = 'Hospital Discharge Medications by Region and Year of Admission';

   ODS proclabel = "Lipid";
   /* SAS procedure code goes here */
   run;
   quit;
ODS PDF close;
```

Figure 3 displays an example of bookmarks that have been customized to the output with the ODS PROCLABEL **=** statement. In this example, each page of the PDF has a table and a graph; each table and graph have a bookmark. The customized bookmarks allow the viewer to more easily navigate the document.

Paper 172-30

Figure 3:



		Year of Discharge								Total	
		2001		2002		2003		2004			
		N	%	N	%	N	%	N	%	N	%
Region	Lipid										
Region 1	No	8	57.14	5	55.56	5	50.00	4	40.00	22	51.16
	Yes	6	42.86	4	44.44	5	50.00	6	60.00	21	48.84
	Total	14	32.56	9	20.93	10	23.26	10	23.26	43	27.39
Region 2	Lipid										
	No	5	55.56	6	42.86	3	30.00	2	28.57	16	40.00
	Yes	4	44.44	8	57.14	7	70.00	5	71.43	24	60.00
	Total	9	22.50	14	35.00	10	25.00	7	17.50	40	25.48
Region 3	Lipid										
	No	7	70.00	5	55.56	7	53.85	3	42.86	22	56.41
	Yes	3	30.00	4	44.44	6	46.15	4	57.14	17	43.59
	Total	10	25.64	9	23.08	13	33.33	7	17.95	39	24.84
Region 4	Lipid										
	No	6	60.00	3	42.86	3	33.33	3	33.33	15	42.86
	Yes	4	40.00	4	57.14	6	66.67	6	66.67	20	57.14
	Total	10	28.57	7	20.00	9	25.71	9	25.71	35	22.29

To prevent SAS from creating bookmarks, use the BOOKMARKLIST= NONE or BOOKMARKLIST=HIDE parameter in the ODS PDF statement. NONE specifies not to generate a list of bookmarks for the PDF file. HIDE generates a list of bookmarks for the PDF file but the bookmarks are not automatically displayed when the file is opened. The code is:

```
ODS PDF file="C:\Projects\SUGI_30\Output\SUGI30_1.PDF"
   author = 'Lori Parsons'
   keywords = 'Regional Discharge Medications Annual'
   subject = 'Process of Care by Region'
   title = 'Hospital Discharge Medications by Region and Year of Admission'
   bookmarklist=none;

   /* SAS procedure code goes here */
   run;
   quit;
ODS PDF close;
```

ODS LAYOUT and ODS REGION

ODS LAYOUT and ODS REGION are new, experimental features in SAS®9. With these features, it is possible to define where on a page the output is to be placed. Multiple regions can be defined in the same layout to position different output on the same page. There are two similar but different ways to implement ODS LAYOUT; the absolute layout and the gridded layout. In absolute layout, you specify the exact position of each region within the layout. In gridded layout, you define columns and fill in “cells” similar to a table. Note, however, that since ODS LAYOUT is considered experimental in SAS®9 (both versions 9.0 and 9.1) the code may change once it is in production.

Paper 172-30

As of version 9.1, the basic code for absolute layout is:

```
ODS LAYOUT START HEIGHT = <value> WIDTH = <value>;
ODS REGION X = <value> Y = <value> HEIGHT = <value> WIDTH = <value>;
   /* SAS procedures go here */
ODS LAYOUT END;
```

For the LAYOUT statement, the entire page is the default size if width and height are not specified. For the REGION statement, the X axis value is measured from the left margin and the y axis value is measured from the top.

PDF File Example 1 (Absolute Layout):

Figure 4 displays an example of combining output from the TABULATE procedure and the GPLOT procedure on the same page by implementing ODS LAYOUT and ODS REGION (absolute layout). This example includes one table and one graph.

The ODS code to create this page is:

```
ODS layout start;

ODS region x=0 in y=.5 in height = 7.5 in width = 5.5 in;
/* PROC TABULATE code goes here */
ODS region x = 5.5 in y = .5 in height = 5 in width = 5.4 in;
/* PROC GPLOT code goes here */

ODS layout end;
```

Figure 4:



		Year of Discharge								Total	
		2001		2002		2003		2004			
		N	%	N	%	N	%	N	%	N	%
Region	Lipid										
Region 1	No	8	57.14	5	55.56	5	50.00	4	40.00	22	51.16
	Yes	6	42.86	4	44.44	5	50.00	6	60.00	21	48.84
	Total	14	32.56	9	20.93	10	23.26	10	23.26	43	27.39
Region 2	Lipid										
	No	5	55.56	6	42.86	3	30.00	2	28.57	16	40.00
	Yes	4	44.44	8	57.14	7	70.00	5	71.43	24	60.00
	Total	9	22.50	14	35.00	10	25.00	7	17.50	40	25.48
Region 3	Lipid										
	No	7	70.00	5	55.56	7	53.85	3	42.86	22	56.41
	Yes	3	30.00	4	44.44	6	46.15	4	57.14	17	43.59
	Total	10	25.64	9	23.08	13	33.33	7	17.95	39	24.84
Region 4	Lipid										
	No	6	60.00	3	42.86	3	33.33	3	33.33	15	42.86
	Yes	4	40.00	4	57.14	6	66.67	6	66.67	20	57.14
	Total	10	28.57	7	20.00	9	25.71	9	25.71	35	22.29

Paper 172-30

PDF File Example 2 (Absolute Layout):

Figure 5 shows another example of displaying PROC TABULATE and PROC GPLOT data. In this example, one table and two graphs are displayed on the same page. The table contains data for all patients. The graphs are subsets of all patients (male patients then female patients). PROC TABLULATE had to be run for each of the graphs to create the appropriate output dataset for PROC GPLOT. We do not want these tables to be included in the PDF file, so we must use the ODS PDF EXCLUDE ALL statement before running PROC TABULATE. We cannot simply add a NOPRINT option to PROC TABULATE or the output dataset would not be created. The ODS PDF select all statement is then used before running PROC GPLOT.

The ODS code (absolute layout) to create this page is:

```
ODS layout start;
ODS region x=0 in y=.5 in height = 7.5 in width = 5.5 in;
/* PROC TABULATE code for the table goes here */

/* Define the region for the male patients PROC GPLOT graph */
ODS region x = 5.5 in y = 0 in height = 3.5 in width = 5.4 in;
ODS PDF exclude all;
/* PROC TABULATE code to get data for PROC GPLOT goes here */
ODS PDF select all;
/* PROC GPLOT code goes here */

/* Define the region for the female patients PROC GPLOT graph */
ODS region x = 5.5 in y = 3.7 in height = 3.5 in width = 5.4 in;
ODS PDF exclude all;
/* PROC TABULATE code to get data for PROC GPLOT goes here */
ODS PDF select all;
/* PROC GPLOT code goes here */
ODS layout end;
```

Figure 5:



Paper 172-30

PDF File Example 3 (Gridded Layout):

In this example, gridded layout is used to produce the same page as in example 2. With gridded layout, the number of columns is defined and ODS makes each region as big as necessary to accommodate the output. This method is simpler than explicitly defining each region, as in absolute layout, however it is more limiting as you cannot define exactly where on the page to place the output. A new region is defined with the ODS region statement; data is filled in left to right and then top to bottom. Figure 6 shows the results of using this method; it is almost identical to example 2. Note that, because many pages of output are produced by the example macro, the ODS PDF STARTPAGE=NOW statement is needed to force the next page.

The ODS code (gridded layout) to create this page is:

```
/* Define the columns of output */
ODS layout start columns=2;

/* Issue an ODS region statement to define the first region (1st column) */
ODS region;

/* PROC TABULATE code for the table goes here */

/* Issue an ODS region statement to define the second region (2nd column) */
ODS region;

/* PROC GPLOT code for Male Patients goes here*/

/* PROC GPLOT code for Female Patients goes here*/

ODS layout end;

ODS PDF startpage = now;
```

Figure 6:



		Year of Discharge								Total	
		2001		2002		2003		2004			
		N	%	N	%	N	%	N	%	N	%
Region	Lipid										
Region 1	No	8	57.14	5	55.56	5	50.00	4	40.00	22	51.16
	Yes	6	42.86	4	44.44	5	50.00	6	60.00	21	48.84
	Total	14	32.56	9	20.93	10	23.26	10	23.26	43	27.39
Region 2	Lipid										
	No	5	55.56	6	42.86	3	30.00	2	28.57	16	40.00
	Yes	4	44.44	8	57.14	7	70.00	5	71.43	24	60.00
	Total	9	22.50	14	35.00	10	25.00	7	17.50	40	25.48
Region 3	Lipid										
	No	7	70.00	5	55.56	7	53.85	3	42.86	22	56.41
	Yes	3	30.00	4	44.44	6	46.15	4	57.14	17	43.59
	Total	10	25.64	9	23.08	13	33.33	7	17.95	39	24.84
Region 4	Lipid										
	No	6	60.00	3	42.86	3	33.33	3	33.33	15	42.86
	Yes	4	40.00	4	57.14	6	66.67	6	66.67	20	57.14
	Total	10	28.57	7	20.00	9	25.71	9	25.71	35	22.29

CONCLUSIONS

The macros presented here contain code to combine tables and graphs into one PDF document, while controlling the look of the final document. These are simple examples of what can be done to combine and organize large amounts of data using the ODS PFD output destination and ODS PDF code.

REFERENCES

Haworth, Lauren E., *Output Delivery System: The Basics,* Cary, NC: SAS Institute Inc., 2001.

Kelley, David. *What's New for PDF Output in the SAS®9 Release of ODS*. http://support.sas.com/news/insider/PDFods.html.

SAS Institute Inc., *New ODS Printer Features for V9* (December 2002). http://support.sas.com/rnd/base/topics/odsprinter/new9.html#LAYOUT.

SAS Institute Inc., *SAS 9.1 Output Delivery System: User's Guide*, Carey, NC: SAS Institute Inc., 2003.

SAS Institute Inc., *TS-659 Exporting SAS/GRAPH Output to PDF Files from Release 8.2 and higher* (March 2004). http://support.sas.com/techsup/technote/ts659/ts659.html

Schellenberger, Brian T., ODS LAYOUT: Arranging ODS Output as You See Fit. http://www2.sas.com/proceedings/sugi28/148-28.PDF

CONTACT INFORMATION

Contact the author at:

Lori S. Parsons
Ovation Research Group
305 Fir Place
Edmonds, WA 98020
Work Phone: (425) 672-8782
Fax: (425) 672-7282
Email: lparsons@ovation.org

Paper 172-30

APPENDIX 1:

```
/* ************************************************************************** */
/* Create dataset used for examples                                           */
/* ************************************************************************** */

proc format;
   value fnoyes
        0 = 'No'
        1 = 'Yes';
   value fRegion
        1 = 'Region 1'
        2 = 'Region 2'
        3 = 'Region 3'
        4 = 'Region 4';
   value fDYear
        1 = '2001'
        2 = '2002'
        3 = '2003'
        4 = '2004';
   value fGender
        1 = 'Male'
        2 = 'Female';
run;

data HospMeds;
   input Region DYear Gender Lipid ASA Beta
         ACE;
   label Region = 'Region';
   label DYear = 'Year of Discharge';
   label Gender = 'Gender';
   label ASA = 'ASA at Discharge';
   label Beta = 'Beta Blocker at Discharge';
   label ACE = 'ACE Inhibitor at Discharge';
   label Lipid = 'Lipid at Discharge';
   format Region fRegion.;
   format DYear fDYear.;
   format Gender fgender.;
   format ASA Beta ACE Lipid fnoyes.;

   datalines;
   1 2001 1 1 1 0 1
   1 2001 2 0 1 1 0
   1 2001 2 0 1 0 1
   1 2001 1 0 0 0 0
   1 2001 2 1 1 1 0
   1 2001 1 0 0 1 0
   1 2001 1 1 1 0 0
   1 2001 2 0 1 1 0
   1 2001 1 0 0 1 0
   1 2001 2 1 0 1 0
   1 2001 1 1 1 0 0
   1 2001 2 0 1 1 0
   1 2001 1 0 0 1 0
   1 2001 2 1 0 1 0
   2 2001 2 1 1 0 1
   2 2001 1 1 1 0 1
   2 2001 1 0 1 1 0
   2 2001 2 0 1 0 1
   2 2001 1 1 1 1 0
   2 2001 2 0 1 0 1
   2 2001 2 1 1 1 0
   2 2001 1 0 1 1 0
   2 2001 2 0 1 0 1
   3 2001 1 1 1 1 0
   3 2001 2 0 0 0 1
   3 2001 1 0 1 1 0
   3 2001 2 0 1 0 1
   3 2001 1 0 0 1 0
   3 2001 2 1 1 0 1
   3 2001 1 0 1 1 0
   3 2001 1 0 0 1 0
   3 2001 2 0 1 0 1
   3 2001 1 1 1 1 0
   4 2001 1 1 1 1 0
   4 2001 2 0 1 0 1
   4 2001 1 1 0 1 0
   4 2001 2 0 1 0 1
   4 2001 1 0 1 1 0
   4 2001 2 1 0 0 1
   4 2001 2 1 1 1 0
   4 2001 2 0 1 0 1
   4 2001 1 0 1 1 0
   4 2001 1 0 1 1 0
   1 2002 1 1 0 0 1
   1 2002 1 0 0 1 0
   1 2002 2 1 1 0 1
   1 2002 1 0 1 1 0
   1 2002 1 1 1 1 0
   1 2002 1 0 1 1 0
   1 2002 2 1 1 1 0
   1 2002 2 0 1 0 1
   1 2002 1 0 1 1 0
   2 2002 2 1 1 0 1
   2 2002 2 1 0 0 1
   2 2002 1 0 1 1 0
   2 2002 2 0 1 0 1
   2 2002 1 1 1 0 1
   2 2002 2 1 0 1 0
   2 2002 2 0 1 0 1
   2 2002 1 1 1 1 0
   2 2002 1 1 1 0 1
   2 2002 2 0 1 0 1
   2 2002 1 0 1 1 0
   3 2002 1 1 0 1 0
   3 2002 2 0 0 0 1
   3 2002 1 0 1 1 0
   3 2002 2 1 1 0 1
   3 2002 1 0 1 1 0
   3 2002 2 0 0 0 1
   3 2002 1 1 1 1 0
   3 2002 2 1 1 0 1
   3 2002 1 0 1 1 0
   4 2002 1 1 1 1 0
   4 2002 2 0 1 0 1
   4 2002 1 0 0 1 0
   4 2002 2 0 1 0 1
   4 2002 1 1 0 1 0
   4 2002 2 1 1 0 1
   4 2002 2 1 1 1 0
   1 2003 1 1 1 0 1
   1 2003 1 1 0 1 0
   1 2003 2 0 0 0 1
   1 2003 1 0 1 1 0
   1 2003 2 1 1 1 0
   1 2003 1 0 1 1 0
   1 2003 1 1 0 1 0
   1 2003 1 0 1 1 0
   1 2003 2 1 0 1 0
   1 2003 1 0 1 1 0
   2 2003 2 1 1 0 1
   2 2003 2 1 1 0 1
   2 2003 1 1 0 1 0
   2 2003 1 0 1 0 1
   2 2003 1 1 1 1 0
   2 2003 2 0 1 0 1
   2 2003 1 1 0 0 1
   2 2003 2 0 1 0 1
   2 2003 1 1 1 1 0
   3 2003 1 1 0 1 0
   3 2003 2 0 0 0 1
   3 2003 1 0 1 1 0
   3 2003 2 1 1 0 1
   3 2003 1 0 1 1 0
   3 2003 1 1 1 1 0
   3 2003 2 0 0 0 1
   3 2003 1 1 1 1 0
   3 2003 2 1 1 0 1
   3 2003 1 0 0 1 0
   3 2003 1 1 1 1 0
   3 2003 2 1 1 0 1
   3 2003 1 0 1 1 0
   3 2003 2 0 1 0 1
   4 2003 1 1 1 1 0
   4 2003 2 0 0 0 1
   4 2003 1 0 1 1 0
   4 2003 2 1 1 0 1
   4 2003 1 1 0 1 0
   4 2003 1 1 1 0 1
   4 2003 1 1 1 1 0
   4 2003 1 0 1 1 0
   4 2003 2 1 1 0 1
   1 2004 1 1 1 0 1
   1 2004 2 1 1 1 0
   1 2004 2 0 0 0 1
   1 2004 1 0 1 1 0
   1 2004 2 1 1 1 0
   1 2004 2 1 0 1 0
   1 2004 1 1 1 1 0
   1 2004 2 0 0 0 1
   1 2004 1 0 1 1 0
   1 2004 1 1 1 1 0
   2 2004 2 1 1 0 1
   2 2004 2 1 1 0 1
   2 2004 1 0 0 1 0
   2 2004 2 1 0 0 1
   2 2004 1 1 0 1 0
   2 2004 2 0 1 0 1
   2 2004 1 1 1 1 0
   3 2004 1 1 0 1 0
   3 2004 2 0 1 0 1
   3 2004 1 0 1 1 0
   3 2004 2 1 0 0 1
   3 2004 1 1 1 1 0
   4 2004 1 1 1 1 0
   4 2004 2 0 1 0 1
   4 2004 1 1 0 1 0
   4 2004 2 1 1 0 1
   4 2004 1 1 1 1 0
   4 2004 2 1 0 0 1
   4 2004 1 0 1 1 0
   4 2004 2 0 0 0 1
   4 2004 2 1 1 1 0
   ;
run;
```

APPENDIX 2:

```
/* ********************************************************************** */
/* Program:        SUGI30_ODS_PDF_1.txt                                   */
/* Subdirectory:   C:\Projects\SUGI_30\SASPrograms                        */
/* Author:         Lori Parsons                                           */
/* Description:    This file contains a macro to run PROC TABULATE and PROC   */
/*                 GPLOT.  One table and one graph will be printed per page.  */
/*                 ODS PDF statements and parameters are used to create a PDF */
/*                 file.  The macro call statements define the outcome        */
/*                 variable, the page title and the bookmark label.           */
/*                                                                        */
/*                 NOTE:  This is an example of Absolute Layout               */
/* ********************************************************************** */


%MACRO MGPlot_Discrete_1
   (outcome,    /* Process/outcome variable name */
    t3,           /* Title of page -- process/outcome variable label */
    t3_short);  /* Shortened Title for bookmark name */

   title1 'Hospital Discharge Medications by Region and Year of Hospital
           Discharge';
   title3 &t3.;

   /* Define the region for the PROC TABULATE data */
   ODS layout start;
   ODS region x=0 in y=.5 in height = 7.5 in width = 5.5 in;

   /* Select data to be output for graph */
   ODS select table;
   ODS output table=outpct;

   /* Define the label for the PDF bookmark */
   ODS proclabel "&t3_short.";

   /* Create the table */
   proc tabulate data=hospmeds format=comma9.0;
      class Region DYear &outcome.;
      table Region * (&outcome. = "&t3_short." all = 'Total'),
            DYear * (n='N' pctn <&outcome. DYear> ='%' * f=7.2) all='Total' *
                    (n='N' pctn <&outcome. Region> ='%' * f=7.2);
      where &outcome. ne .;
   run;
   quit;
   ODS output close;

   /* Keep the column percents for the "YES" response */
   /* This will be the file used by PROC GPLOT */
   data graphit (keep = out_pct Region DYear);
     set outpct (rename=(PctN_110 = out_pct));
     if &outcome. = 1;
   run;

   /* Plot the data */
   goptions reset=all rotate=landscape;

   /* Define the region for the PROC GPLOT graph */
   ODS region x = 5.5 in y = .5 in height = 5 in width = 5.4 in;
```

```
   /* Define the label for the PDF bookmark */
   ODS proclabel "PLOT:  &t3_short.";

   proc gplot data=graphit;
      plot out_pct * DYear = Region / haxis = axis1 vaxis = axis2;
      symbol1 v=dot       i=join l=1 w=5 h=1.5 c=red;
      symbol2 v=triangle  i=join l=1 w=5 h=1.5 c=blue;
      symbol3 v=circle    i=join l=2 w=5 h=1.5 c=green;
      symbol4 v=square    i=join l=2 w=5 h=1.5 c=black;

      axis2 label = (j=center "%" a=90 f=CENTB)
            length = 4.8 in
            order =  25 to 100 by 5;
      axis1 label = (f=CENTB)
            length = 8 in;
   run;
   quit;

   ODS layout end;

%Mend MGPlot_Discrete_1;


/* ********************************************************** */
/* Create the PDF file with ODS PDF statements                */
/* Call the macro MGPlot_Discrete_1 one time for each outcome */
/* ********************************************************** */

options nocenter nodate orientation=landscape;
ODS PDF file="C:\Projects\SUGI_30\Output\SUGI30_1.PDF"
      author = 'Lori Parsons'
      keywords = 'Regional Discharge Medications Annual'
      subject = 'Process of Care by Region'
      title = 'Hospital Discharge Medications by Region and Year of Admission';

   %MGPlot_Discrete_1 (Lipid, Discharged on Lipid Lowering Agent, Lipid);
   %MGPlot_Discrete_1 (ASA, Discharged on ASA, ASA);
   %MGPlot_Discrete_1 (Beta, Discharged on Beta Blocker, Beta Blocker);
   %MGPlot_Discrete_1 (ACE, Discharged on ACE Inhibitor, ACE Inhibitor);
ODS PDF close;
```

APPENDIX 3:

```
/* ********************************************************************** */
/* Program:        SUGI30_ODS_PDF_2.txt                                   */
/* Subdirectory:   C:\Projects\SUGI_30\SASPrograms                        */
/* Author:         Lori Parsons                                           */
/* Description:    This file contains two macros to run PROC TABULATE and PROC*/
/*                 GPLOT.  One table and two graphs will be printed per page. */
/*                 ODS PDF statements and parameters are used to create a PDF */
/*                 file.  The macro call statements define the outcome    */
/*                 variable, the page title and the bookmark label.       */
/*                                                                        */
/*                 NOTE:  This is an example of Absolute Layout           */
/* ********************************************************************** */
```

Paper 172-30

```
/* ********************************************************** */
/* Macro MGraphit is used to create a graph for a subpopulation */
/* Call this macro once for each graph of each outcome          */
/* ********************************************************** */
%MACRO MGraphit (subpopl,subtitle,bmtitle);
   /* Select data to be output for graph */
   ODS select table;
   ODS output table=outpct;

  /* Create the output file, but do not send the table to the PDF file */
  /* NOTE: The NOPRINT option cannot be used in PROC TABULATE or the output file */
  /* will not be created.  Therefore, the ODS PDF EXCLUDE ALL statement was used */

   ODS PDF exclude all;
   proc tabulate data=hospmeds;
      class Region DYear &outcome.;
      table Region * (&outcome. = "&t3_short." all = 'Total'),
            DYear * (n='N' pctn <&outcome. DYear> ='%' * f=7.2) all='Total' *
           (n='N' pctn <&outcome. Region> ='%' * f=7.2);
         where &outcome. ne . and &subpopl.;
   run;
   ODS output close;

   ODS PDF select all;

   /* Keep the column percents for the "YES" response */
   /* This will be the file used by PROC GPLOT */
   data graphit (keep = out_pct Region DYear);
     set outpct (rename=(PctN_110 = out_pct));
     if &outcome. = 1;
   run;

   /* Plot the data */
   goptions reset=all rotate=landscape;

   /* Define the label for the PDF bookmark */
   ods proclabel "PLOT:  &t3_short., &bmtitle.";

   proc gplot data=graphit;
      plot out_pct * DYear = Region / haxis = axis1 vaxis = axis2;
      symbol1 v=dot       i=join l=1 w=5 h=1.5 c=red;
      symbol2 v=triangle  i=join l=1 w=5 h=1.5 c=blue;
      symbol3 v=circle    i=join l=2 w=5 h=1.5 c=green;
      symbol4 v=square    i=join l=2 w=5 h=1.5 c=black;

      axis2 label = (j=center "%" a=90 f=CENTB) length = 4.8 in order = 25 to 100
            by 5;
      axis1 label = (f=CENTB) length = 8 in;

      title1 f=centb "&subtitle.";
   run;
   quit;

%Mend MGraphit;

/* ********************************************************** */
/* Create the table for all patients and send to the PDF file */
/* Call the macro MGraphit one time for male patients and one */
/* time for female patients for each outcome                  */
/* ********************************************************** */
```

```
%MACRO MGPlot_Discrete_2
   (outcome,    /* Process/outcome variable name */
    t3,           /* Title of page -- process/outcome variable label */
    t3_short);  /* Shortened Title for bookmark name */

   title1 'Hospital Discharge Medications by Region and Year of Hospital
          Discharge';
   title3 &t3.;

   /* Define the region for the PROC TABULATE data */
   ODS layout start;
   ODS region x=0 in y=.5 in height = 7.5 in width = 5.5 in;

   ODS proclabel "&outcome.";

   /* Create the table for all patients */
   proc tabulate data=hospmeds format=comma9.0;
      class Region DYear &outcome.;
      table Region * (&outcome. = "&t3_short." all = 'Total'),
            DYear * (n='N' pctn <&outcome. DYear> ='%' * f=7.2) all='Total' *
            (n='N' pctn <&outcome. Region> ='%' * f=7.2);
      where &outcome. ne .;
   run;
   quit;


   /* MALES ONLY GRAPH */
   /* Define the region for the MALE PATIENTS PROC GPLOT graph */
   ODS region x = 5.5 in y = 0 in height = 3.5 in width = 5.4 in;
   /* Get the graph for the MALE PATIENTS */
   %MGraphit(gender eq 1,Male Patients,Males);

   /* FEMALES ONLY GRAPH */
   /* Define the region for the FEMALE PATIENTS PROC GPLOT graph */
   ODS region x = 5.5 in y = 3.7 in height = 3.5 in width = 5.4 in;
   /* Get the graph for the FEMALE PATIENTS */
   %MGraphit(gender eq 2,Female Patients,Females);

   ODS layout end;

%Mend MGPlot_Discrete_2;


/* ********************************************************** */
/* Create the PDF file with ODS PDF statements                */
/* Call the macro MGPlot_Discrete_2 one time for each outcome */
/* ********************************************************** */

options nocenter nodate orientation=landscape;
ODS PDF file="C:\Projects\SUGI_30\Output\SUGI30_2.PDF"
      author = 'Lori Parsons'
      keywords = 'Regional Discharge Medications Annual'
      subject = 'Process of Care by Region'
      title = 'Hospital Discharge Medications by Region and Year of Admission';

   %MGPlot_Discrete_2 (Lipid, Discharged on Lipid Lowering Agent, Lipid);
   %MGPlot_Discrete_2 (ASA, Discharged on ASA, ASA);
   %MGPlot_Discrete_2 (Beta, Discharged on Beta Blocker, Beta Blocker);
   %MGPlot_Discrete_2 (ACE, Discharged on ACE Inhibitor, ACE Inhibitor);
ODS PDF close;
```

APPENDIX 4:

```
/* ********************************************************************** */
/* Program:        SUGI30_ODS_PDF_3.txt                                   */
/* Subdirectory:   C:\Projects\SUGI_30\SASPrograms                        */
/* Author:         Lori Parsons                                           */
/* Description:    This file contains two macros to run PROC TABULATE and PROC*/
/*                 GPLOT.  One table and two graphs will be printed per page. */
/*                 ODS PDF statements and parameters are used to create a PDF */
/*                 file.  The macro call statements define the outcome    */
/*                 variable, the page title and the bookmark label.       */
/*                                                                        */
/*                 NOTE:  This is an example of Gridded Layout            */
/* ********************************************************************** */

/* ********************************************************** */
/* Macro MGraphit is used to create a graph for a subpopulation */
/* Call this macro once for each graph of each outcome        */
/* ********************************************************** */

%MACRO MGraphit (subpopl,subtitle,bmtitle);

   /* Select data to be output for graph */
   ODS select table;
   ODS output table=outpct;

  /* Create the output file, but do not send the table to the PDF file */
  /* NOTE: The NOPRINT option cannot be used in PROC TABULATE or the output file */
  /* will not be created.  Therefore, the ODS PDF EXCLUDE ALL statement was used */

   ods pdf exclude all;

   proc tabulate data=hospmeds;
      class Region DYear &outcome.;
      table Region * (&outcome. = "&t3_short." all = 'Total'),
            DYear * (n='N' pctn <&outcome. DYear> ='%' * f=7.2) all='Total' *
            (n='N' pctn <&outcome. Region> ='%' * f=7.2);
       where &outcome. ne . and &subpopl.;
   run;
   ods output close;

   ods pdf select all;

   /* Keep the column percents for the "YES" response */
   /* This will be the file used by PROC GPLOT */
   data graphit (keep = out_pct Region DYear);
     set outpct (rename=(PctN_110 = out_pct));
     if &outcome. = 1;
   run;

   /* Plot the data */
   goptions reset=all rotate=landscape hsize = 7.7 vsize = 4.5;

   /* Define the label for the PDF bookmark */
   ods proclabel "PLOT:  &t3_short., &bmtitle.";

   proc gplot data=graphit;
      plot out_pct * DYear = Region / haxis = axis1 vaxis = axis2;
      symbol1 v=dot       i=join l=1 w=5 h=1.5 c=red;
      symbol2 v=triangle  i=join l=1 w=5 h=1.5 c=blue;
```

```
      symbol3 v=circle    i=join l=2 w=5 h=1.5 c=green;
      symbol4 v=square    i=join l=2 w=5 h=1.5 c=black;

      axis2 label = (j=center "%" a=90 f=CENTB)    order =  25 to 100 by 5;
      axis1 label = (f=CENTB)   ;

      title1 f=centb "&subtitle.";
   run;
   quit;
%Mend MGraphit;

/* ********************************************************** */
/* Create the table for all patients and send to the PDF file */
/* Call the macro MGraphit one time for male patients and one */
/* time for female patients for each outcome                  */
/* ********************************************************** */

%MACRO MGPlot_Discrete_3
   (outcome,    /* Process/outcome variable name */
   t3,            /* Title of page -- process/outcome variable label */
   t3_short);  /* Shortened Title for bookmark name */

   title1 'Hospital Discharge Medications by Region and Year';
   title3 &t3.;

   /* Define the columns of output */
   ODS layout start columns=2;

   /* Issue an ODS region statement to define the first region (1st column) */
   ODS region;

   ODS proclabel "&outcome.";

   /* Create the table for all patients */
   proc tabulate data=hospmeds format=comma9.0;
      class Region DYear &outcome.;
      table Region * (&outcome. = "&t3_short." all = 'Total'),
            DYear * (n='N' pctn <&outcome. DYear> ='%' * f=7.2) all='Total' *
            (n='N' pctn <&outcome. Region> ='%' * f=7.2);
      where &outcome. ne .;
   run;
   quit;

   /* Issue an ODS region statement to define the second region (2nd column) */
   ODS region;

   /* MALES ONLY GRAPH */
   /* Get the graph for the MALE PATIENTS */
   %MGraphit(gender eq 1,Male Patients,Male);

   /* FEMALES ONLY GRAPH */
   /* Get the graph for the FEMALE PATIENTS */
   %MGraphit(gender eq 2,Female Patients,Female);

   ODS layout end;

   ODS pdf startpage = now;

%Mend MGPlot_Discrete_3;
```

Paper 172-30

```
/* ********************************************************** */
/* Create the PDF file with ODS PDF statements                */
/* Call the macro MGPlot_Discrete_2 one time for each outcome */
/* ********************************************************** */

/* ************** */
options nocenter nodate orientation=landscape;
ODS pdf FILE="C:\Projects\SUGI_30\Output\SUGI30_3.pdf"
      author = 'Lori Parsons'
      keywords = 'Regional Discharge Medications Annual'
      subject = 'Process of Care by Region'
      title = 'Hospital Discharge Medications by Region and Year';

   %MGPlot_Discrete_3 (Lipid, Discharged on Lipid Lowering Agent, Lipid);
   %MGPlot_Discrete_3 (ASA, Discharged on ASA, ASA);
   %MGPlot_Discrete_3 (Beta, Discharged on Beta Blocker, Beta Blocker);
   %MGPlot_Discrete_3 (ACE, Discharged on ACE Inhibitor, ACE Inhibitor);
ODS pdf close;
```